BIOM 505: Biostatistical Methods I (Fall 2018)

Lab 3

CALCULATING BINOMIAL AND NORMAL PROBABILITIES USING SAS

Dr. Fares Qeadan: fqeadan@salud.unm.edu

REMARK:

Let X~Bin(n, p) with xϵ{0,1,2,3,…,n} then P(X=x)= PDF('binomial', x, p, n) by SAS

Let X~Bin(n, p) with xϵ{0,1,2,3,…,n} then P(X≤x)=probbnml(p, n, x) by SAS

Let X~Bin(n, p) with xϵ{0,1,2,3,…,n} then P(X≤x)=CDF('binomial', x, p, n) by SAS

Let X~Bin(n, p) with xϵ{0,1,2,3,…,n} then P(X>x)=SDF('binomial', x, p, n) by SAS

QUESTION-1:

The prevalence of Lower Back Pain (LBP) among adults in the USA is about 65%. Suppose we obtain a random sample of 20 adults, then the random variable X denoting adults from the sample is binomial with n=20 and p=0.65.

1. What is the probability of having exactly 13 adults with LBP? P (X = 13)
2. What is the probability of having at most 13 adults with LBP? P (X ≤ 13)
3. What is the probability of having at least 13 adults with LBP? P (X ≥ 13)
4. What is the probability of having more than 13 adults with LBP? P (X > 13)
5. What is the probability of having less than 13 adults with LBP? P (X < 13)
6. What is the probability of having more than 7 and less than 11 adults with LBP? P(8 ≤ X ≤ 10)

SOLUTION:

Part (1): P (X = 13)

```
data Q1_1;
prob1_a= PDF('binomial',13,0.65,20);
prob1_b = probbnml(0.65, 20, 13)- probbnml(0.65, 20, 12);
prob1_c= CDF('binomial', 13, 0.65, 20)- CDF('binomial', 12, 0.65, 20);
run;

proc print data=Q1_1;
run;
```

Obs	prob1_a	prob1_b	prob1_c
1	0.18440	0.18440	0.18440

Part (2): P (X ≤ 13)

```
data Q1_2;
prob2_a = probbnml(0.65, 20, 13);
prob2_b= CDF('binomial', 13, 0.65, 20);
run;

proc print data=Q1_2;
run;
```

Obs	prob2_a	prob2_b
1	0.58337	0.58337

Part (3): P (X ≥ 13)

```
data Q1_3;
prob3_a = 1 - probbnml(0.65, 20, 12);
prob3_b= 1- CDF('binomial', 12, 0.65, 20);
prob3_c=SDF('binomial',12,.65,20);
run;

proc print data=Q1_3;
run;
```

Obs	prob3_a	prob3_b	prob3_c
1	0.60103	0.60103	0.60103

Part (4): P (X > 13)

```
data Q1_4;
prob4_a = 1 - probbnml(0.65, 20, 13);
prob4_b= 1- CDF('binomial', 13, 0.65, 20);
prob4_c=SDF('binomial',13,.65,20);
run;

proc print data=Q1_4;
run;
```

Obs	prob4_a	prob4_b	prob43_c
1	0.41663	0.41663	0.41663

Part (5): P (X < 13)

```
data Q1_5;
prob5_a = probbnml(0.65, 20, 12);
prob5_b= CDF('binomial', 12, 0.65, 20);
run;

proc print data=Q1_5;
run;
```

Obs	prob5_a	prob5_b
1	0.39897	0.39897

Part (6): P (8 ≤ X ≤ 10

```
data Q1_6;
prob6_a= PDF('binomial',8,0.65,20)+ PDF('binomial',9,0.65,20)+
PDF('binomial',10,0.65,20);
prob6_b = probbnml(0.65, 20, 10)- probbnml(0.65, 20, 7);
prob6_c= CDF('binomial', 10, 0.65, 20)- CDF('binomial', 7, 0.65, 20);
run;

proc print data=Q1_6;
run;
```

Obs	prob6_a	prob6_b	prob6_c
1	0.11577	0.11577	0.11577

REMARK:

Let Z~N(0, 1) then P(Z≤z)= probnorm(z) OR P(Z≤z)= CDF('normal', z) by SAS

Let X~N(μ, σ) then P(X≤x)= probnorm$\left(\frac{x-\mu}{\sigma}\right)$ OR P(X≤x)= CDF('normal', x, μ, σ) by SAS

Let X~N(μ, σ) then P(X>x)= SDF('normal', x, μ, σ) by SAS

Let Z~N(0, 1) such that p=Pr(Z≤z*) then z*=probit(p) OR z*=QUANTILE('Normal', p, 0, 1)

Let X~N(μ, σ) such that p=Pr(X≤x*) then x*=μ+probit(p)*σ OR x*=QUANTILE('Normal', p, μ, σ)

QUESTION-2:

Let X equal the IQ of a randomly selected adult in the USA. Assume X ~ N(100, 15).

1. What is the chance that a randomly selected adult has an IQ of 90? P(X=90)
2. What is the chance that a randomly selected adult has an IQ below 90? P(X<90)
3. What is the chance that a randomly selected adult has at most an IQ of 90? P(X≤90)
4. What is the chance that a randomly selected adult has an IQ over 90? P(X>90)
5. What is the chance that a randomly selected adult has an IQ between 90 and 110? P(90<X<110)
6. An IQ above what puts you in the top 10%?
7. Which IQ puts you in the lower 10%?

SOLUTION:

Part (1): P(X=90)

SAS is not required! The answer is zero as **P(X=c) = 0** for continuous random variables.

Part (2): P(X<90)

```
data Q2_2;
prob2_a= probnorm((90-100)/15);
prob2_b =CDF('normal', 90, 100, 15);
prob2_c =1-SDF('normal', 90, 100, 15);
run;
```

```
proc print data=Q2_2;
run;
```

Obs	prob2_a	prob2_b	prob2_c
1	0.25249	0.25249	0.25249

Part (3): P(X≤90)

```
data Q2_3;
prob3_a= probnorm((90-100)/15);
prob3_b =CDF('normal', 90, 100, 15);
prob3_c =1-SDF('normal', 90, 100, 15);
run;

proc print data=Q2_3;
run;
```

Obs	prob2_a	prob2_b	prob2_c
1	0.25249	0.25249	0.25249

Part (4): P(X>90)

```
data Q2_4;
prob4_a= 1-probnorm((90-100)/15);
prob4_b =1-CDF('normal', 90, 100, 15);
prob4_c =SDF('normal', 90, 100, 15);
run;

proc print data=Q2_4;
run;
```

Obs	prob4_a	prob4_b	prob4_c
1	0.74751	0.74751	0.74751

Part (5): P(90<X<110)

```
data Q2_5;
prob5_a= probnorm((110-100)/15)-probnorm((90-100)/15);
prob5_b = CDF('normal', 110, 100, 15)-CDF('normal', 90, 100, 15);
prob5_c =SDF('normal', 90, 100, 15)- SDF('normal', 110, 100, 15);
run;

proc print data=Q2_5;
run;
```

Obs	prob5_a	prob5_b	prob5_c
1	0.49501	0.49501	0.49501

Part (6): 0.90=P(X<x*) OR 0.10=P(X>x*)

```
data Q2_6;
xstar_a= 100+probit(0.90)*15;
xstar_b = QUANTILE('Normal', 0.90, 100, 15);
run;

proc print data=Q2_6;
run;
```

Obs	xstar_a	xstar_b
1	119.223	119.223

Part (7): 0.10=P(X<x*) OR 0.90=P(X>x*)

```
data Q2_7;
xstar_a= 100+probit(0.10)*15;
xstar_b = QUANTILE('Normal', 0.10, 100, 15);
run;

proc print data=Q2_7;
run;
```

Obs	xstar_a	xstar_b
1	80.7767	80.7767